Supplement:

Ecological Public Goods Games: Cooperation and Bifurcations

Christoph Hauert[1] , Joe Yuishiro Wakano[2] & Michael Doebeli[3]

[1] Program for Evolutionary Dynamics, Harvard University, One Brattle Square, Cambridge, MA 02138, USA
[2] Departments of Biological Sciences, University of Tokyo, 7-3-1 Hongo, Bunkyou-ku, Tokyo, 113-0033, Japan
[3] Department of Zoology and Department of Mathematics, UBC,
6270 University Boulevard, Vancouver B.C. Canada, V6T 1Z4

Definitions

```
JacobianMatrix[f_List?VectorQ, x_List] :=
 Outer[D, f, x] /; Equal @@ (Dimensions /@ {f, x})
```

Equations of Motion (b=0)

```
G[q_, z_] := {-q z (1 - q) F[z], -(1 - z) (z q (r - 1) (1 - z^(n-1)) - d)}
F[z_] := 1 + (r - 1) z^(n-1) - r (1 - z^n)/(n (1 - z))
```

$$G[q, z] = \{-q\,z\,(1-q)\,F[z],\ -(1-z)\,(z\,q\,(r-1)\,(1-z^{n-1})-d)\}$$

$$F[z] = 1 + (r-1)\,z^{n-1} - \frac{r\,(1-z^n)}{n\,(1-z)}$$

Equilibrium Q

Multiplication factor *r* at equilibrium

```
FullSimplify[Solve[{F[z] == 0}, r]]
```

$$\left\{\left\{r \to \frac{n\,(-1+z)\,(-z+z^n)}{z+(n\,(-1+z)-z)\,z^n}\right\}\right\}$$

One coordinate of **Q** :

```
Simplify[Solve[{G[q, z][[2]] == 0}, q]]
```

$$\left\{\left\{q \to -\frac{d}{(-1+r)\,(-z+z^n)}\right\}\right\}$$

Jacobian at Q

Clear $F[z]$ to keep things simple

```
F[z_] =.
```

```
J = FullSimplify[JacobianMatrix[G[q, z], {q, z}] /. {F[z] → 0}];
J // MatrixForm
```

$$\begin{pmatrix} 0 & (-1+q)\, q\, z\, F'[z] \\ -(-1+r)\,(-1+z)\,(-z+z^n) & \frac{-d\, z+q\,(-1+r)\,(-z^n\,(n\,(-1+z)+z)+z\,(-1+2\,z))}{z} \end{pmatrix}$$

The trace of the Jacobian vanishes at the Hopf bifurcation. This condition yields the other coordinate of **Q** :

```
Solve[FullSimplify[
   Tr[J] //. {r → n (-1 + z) (-z + z^n) / (z + (n (-1 + z) - z) z^n), q → - d / ((-1 + r) (-z + z^n))}] == 0, z]
```

$$\left\{\left\{z \to n^{\frac{1}{1-n}}\right\}\right\}$$

Coordinates of **Q** and bifurcation point expressed in terms of the group size N and the per capita death rate d:

```
hopfparams := {z → n^(-1/(n-1)), q → d n^(n/(n-1)) / ((n - 1) (n^(n/(n-1)) - n - 1)), r → n (n^(1/(n-1)) - 1)}
```

- **Verify:**

Define $F[z]$ in order to verify that **Q** is indeed an equilibrium ($G[q,\ z] = 0$) and that the Trace[J] vanishes.

```
F[z_] := 1 + (r - 1) z^(n-1) - r (1 - z^n) / (n (1 - z));
```

```
FullSimplify[G[q, z] /. hopfparams, n > 2]
```

{0, 0}

```
FullSimplify[Tr[J] /. hopfparams, n > 2]
```

0

Clear $F[z]$

```
F[z_] =.
```

Analysis of Hopf bifurcation

Compact representation of Jacobian at Hopf bifurcation

```
J = ( 0              -(1 - q) q z F′[z] )
    ( -d (1-z)/q     0                  ); J // MatrixForm
```

$$\begin{pmatrix} 0 & (-1+q)\, q\, z\, F'[z] \\ -\frac{d\,(1-z)}{q} & 0 \end{pmatrix}$$

- **Verify:**

```
FullSimplify[JacobianMatrix[G[q, z], {q, z}] - J /. F[z] → 0 //. hopfparams,
  n > 2] // MatrixForm
```

$$\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

Eigenvalues & Eigenvectors of Jacobian J

At the Hopf bifurcation the Jacobian has two purely imaginary eigenvalues $\pm\, i\, \omega$:

```
Simplify[Det[J]]
```

d z (-1 + q + z - q z) F′[z]

```
ω := Sqrt[-d z (1 - q) (1 - z) F′[z]]
```

The eigenvectors must be chosen such that they satisfy the normalization $\langle p_1,\ p_2\rangle = 1$ where $\langle p,\ q\rangle = \overline{p}_1 \cdot q_1 + \overline{p}_2 \cdot q_2$ denotes the standard scalar product in $\mathbb{C}^2$:

```
p1 := {1, ⅈ d (1 - z)/(q ω)}
p2 := 1/2 {1, ⅈ ω q/(d (1 - z))}
```

- **Verify:**

```
Simplify[J.p1 - ⅈ ω p1]
{0, 0}
```

```
Simplify[Transpose[J].p2 + ⅈ ω p2]
{0, 0}
```

```
FullSimplify[Conjugate[p2].p1,
 c > 0 && d > 0 && ω > 0 && r > 0 && n > 2 && q > 0 && z > 0]
1
```

Clear ω:

```
ω =.
```

Change of coordinates

Translation of coordinates such that **Q** is in origin

```
y[v_, v1_] := {q, z} + v p1 + v1 Conjugate[p1];
```

The function $H[y,\ p]$ essentially determines the change in distance between y and **Q**.

```
H[y_, p2_] := Conjugate[p2].G[y[[1]], y[[2]]];
```

Taylor expansion of *H*[*y*, *p*] at origin (i.e. at Q)

The first Lyapunov coefficient l_1 is determined by three coefficients of this expansion, g_{20}, g_{11} and g_{21}.
In order to keep the expresions simple, some substitutions regarding the Hopf bifurcation are made (using $F[z] = 0$ and $z^N = z/N$).

```
g20 = FullSimplify[
  Simplify[D[H[y[v, v1], p2], v, v] /. {v → 0, v1 → 0}] /. {F[z] → 0, z^n → z/n},
  d > 0 && r > 0 && ω > 0 && n > 2 && q > 0 && z > 0 && z < 1]
```

$$\frac{1}{2\,q\,\omega^2}\left(d\,(-1+z)\left(\frac{(-1+n)\,q\,(-1+r)\,\omega\,(\mathrm{i}\,d\,n\,(-1+z)^2+2\,z^2\,\omega)}{d\,n\,(-1+z)\,z}-2\,(d\,(-1+q)\,(-1+z)+\mathrm{i}\,(-1+2\,q)\,z\,\omega)\,F'[z]+d\,z\,(-1+q+z-q\,z)\,F''[z]\right)\right)$$

```
g11 = FullSimplify[Simplify[D[H[y[v, v1], p2], v, v1] /. {v → 0, v1 → 0}] /.
    {F[z] → 0, z^n → z/n}, d > 0 && r > 0 && ω > 0 && n > 2 && q > 0 && z > 0]
```

$$\frac{d\,(-1+z)^2\,(-i\,(-1+n)\,q\,(-1+r)\,\omega+d\,(-1+q)\,z\,(2\,F'[z]+z\,F''[z]))}{2\,q\,z\,\omega^2}$$

```
g21 =
 FullSimplify[Simplify[D[H[y[v, v1], p2], v, v, v1] /. {v → 0, v1 → 0}] /.
    {F[z] → 0, z^n → z/n}, d > 0 && r > 0 && ω > 0 && n > 2 && q > 0 && z > 0 && z < 1]
```

$$\frac{1}{2\,q^2\,\omega^3}\Big(d\,(-1+z)\Big(\frac{1}{z^3}((-1+n)\,q\,(-1+r)\,(-1+z)\,\omega\,(-3\,d\,z^2-n\,z^n\,(d\,(-2+n)\,(-1+z)+i\,z\,\omega)))+2\,\omega\,(d\,(-1+2\,q)\,(-1+z)-i\,q\,z\,\omega)\,F'[z]+d\,(-1+z)\,((-3\,i\,d\,(-1+q)\,(-1+z)+(-1+2\,q)\,z\,\omega)\,F''[z]-i\,d\,(-1+q)\,(-1+z)\,z\,F^{(3)}[z])\Big)\Big)$$

Determine imaginary part of $g_{20}\,g_{11}$.

```
img20g11 = FullSimplify[Im[ComplexExpand[g20 g11]], d > 0 && r > 2 && ω > 0 &&
    n > 2 && q > 0 && q < 1 && z > 0 && z < 1 && F'[z] ∈ Reals && F''[z] ∈ Reals]
```

$$-\frac{1}{2\,n\,q^2\,z\,\omega^3}\,(d\,(-1+z)^2\,((-1+n)^2\,q^2\,(-1+r)^2\,z\,\omega^2+d^2\,n\,(-1+q)\,(-1+z)\,(-(-1+n)\,q\,(-1+r)\,(-1+z)+(-1+2\,q)\,z^2\,F'[z])\,(2\,F'[z]+z\,F''[z])))$$

Similarly, determine real part of g_{21}.

```
reg21 = Simplify[Re[ComplexExpand[g21]],
   d > 0 && r > 2 && ω > 0 && n > 2 && q > 0 && q < 1 && z > 0 &&
    z < 1 && F'[z] ∈ Reals && F''[z] ∈ Reals && F'''[z] ∈ Reals]
```

$$\frac{1}{2\,q^2\,z^3\,\omega^2}\,(d^2\,(-1+z)^2\,(-(-1+n)\,q\,(-1+r)\,(3\,z^2-(-2+n)\,n\,z^n+(-2+n)\,n\,z^{1+n})+2\,(-1+2\,q)\,z^3\,F'[z]+(-1+2\,q)\,z^4\,F''[z]))$$

Intermezzo: Determine $F'[z]$ and $F''[z]$ at the Hopf bifurcation

```
F[z_] := 1 + (r - 1) z^(n-1) - r (1 - z^n) / (n (1 - z))
```

Use $F[z] = 0$ and $z = n^{-\frac{1}{n-1}}$:

- ***F*'[*z*]:**

```
fpz := (n - 1)/n ((r (1 - z) - 1)/(z (1 - z)))
```

- **Verify:**

```
FullSimplify[F'[z] - fpz /. hopfparams, n > 2]
0
```

- ***F*''[*z*]:**

```
fppz := (n - 1) ((n - 2)/(n z^2) (r - 1) + r/(n z (1 - z)) - 2/(n (1 - z)^2))
```

- **Verify:**

```
FullSimplify[F''[z] - fppz /. hopfparams, n > 2]
0
```

Clear $F[z]$ - no longer needed

```
F[z_] =.
```

First Lyapunov Coefficient l_1

```
l1 = FullSimplify[1/(2 ω^2) (-img20g11 + ω reg21), d > 0 && r > 2 && ω > 0 &&
    n > 2 && q > 0 && q < 1 && z > 0 && z < 1 && F'[z] ∈ Reals && F''[z] ∈ Reals]
```

$$\frac{1}{4\,q^2\,z^3\,\omega^5}\left(d\,(-1+z)^2\left(\frac{1}{n}((-1+n)\,q\,(-1+r)\right.\right.$$
$$((-1+n)\,q\,(-1+r)\,z^3+d\,n\,(-3\,z^2-(-2+n)\,n\,(-1+z)\,z^n))\,\omega^2)+$$
$$d\,z^2\,(-d\,(-1+n)\,(-1+q)\,q\,(-1+r)\,(-1+z)^2+(-1+2\,q)\,z\,\omega^2+$$
$$\left.\left.d\,(-1+q)\,(-1+2\,q)\,(-1+z)\,z^2\,F'[z])\,(2\,F'[z]+z\,F''[z])\right)\right)$$

Simplify l_1 :

```
FullSimplify[l1 /. {ω^2 → -d z (1 - q) (1 - z) F′[z], z^n → z/n}]
```

$$\frac{1}{4\,n\,q\,z\,\omega^5}\left(d^2\ (-1+n)\ (-1+q)\ (-1+r)\ (-1+z)^3\ \left(\left(q\ (-1+n+r-n\,r)\ z^2+d\,n\ (4-n+(-1+n)\ z)\right) F'[z]-d\,n\ (-1+z)\ z\,F''[z]\right)\right)$$

The first factor is:

$$\frac{d^2\ (n-1)\ (1-q)\ (r-1)\ (1-z)^3}{4\,n\,q\,z\,\omega^5}$$

and the second factor is:

$$(q\ (-1+n+r-n\,r)\ z^2+d\,n\ (4-n+(-1+n)\ z))\ F'[z]-d\,n\ (-1+z)\ z\,F''[z]$$

Simplify the second factor:

```
FullSimplify[
 (q (-1 + n + r - n r) z^2 + d n (4 - n + (-1 + n) z)) F′[z] - d n (-1 + z) z F″[z]
  //. {F'[z] → fpz, F''[z] → fppz, q → d n/((r - 1) z (n - 1)), r → n (1 - z)/z}
]
```

$$-\frac{d\ (-1+n)\ (n\ (-2+z)\ (-1+z)-2\,z\ (1+z))}{(-1+z)\ z^2}$$

With this, the first factor becomes:

$$\frac{d^3\ (1-q)\ (n-1)^2\ (r-1)\ (1-z)^2}{4\,q\,n\,z^3\,\omega^5}$$

and is always positive. The second factor reduces to:

$$n\ (2-z)\ (1-z)-2\,z\ (1+z)$$

and determines the roots of l_1 .

- **Verify:**

```
FullSimplify[l1 -
    (d^3 (1 - q) (n - 1)^2 (r - 1) (1 - z)^2 / (4 q n z^3 ω^5)) (n (2 - z) (1 - z) - 2 z (1 + z)) /.
   {F'[z] → fpz, F''[z] → fppz, ω^2 → -d z (1 - q) (1 - z) fpz} //.
  hopfparams, n > 2]
```

0

Roots of l_1

```
FullSimplify[n (2 - z) (1 - z) - 2 z (1 + z) /. hopfparams, n > 2]
```

$n^{-\frac{2}{-1+n}} \left(-2 + n - 2\, n^{\frac{1}{-1+n}} + n^{\frac{n}{-1+n}} \left(-3 + 2\, n^{\frac{1}{-1+n}}\right)\right)$

Note that this expression only depends on the group size N and is independent of d.

```
FindRoot[-2 + n - 2 n^(1/(-1+n)) + n^(n/(-1+n)) (-3 + 2 n^(1/(-1+n))) == 0, {n, 2}]
```

{n → 2.}

```
FindRoot[-2 + n - 2 n^(1/(-1+n)) + n^(n/(-1+n)) (-3 + 2 n^(1/(-1+n))) == 0, {n, 10}]
```

{n → 8.49298}

l_1 has two roots, one at $N = 2$ and the other at $N = 8.493$.
The slope at $N = 2$ is negative:

```
Simplify[D[-2 + n - 2 n^(1/(-1+n)) + n^(n/(-1+n)) (-3 + 2 n^(1/(-1+n))), n] /. n → 2]
```

11 - 16 Log[2]

and thus $l_1 < 0$ holds, i.e. the Hopf bifurcation is supercritical, for $2 < N < 8.493$. For $N > 8.493$ the Hopf bifurcation is subcritical ($l_1 > 1$).

References

Kuznetsov, Y. (2004) Elements of Applied Bifurcation Theory, Springer, p.89ff

Kuznetsov, Y. (2006) Andronov-Hopf Bifurcation on Scholarpedia: http://www.scholarpedia.org/article/Andronov-Hopf_Bifurcation